(define (exponentiation? x)
  (and (pair? x) (eq? (car x) '**)))

(define (base s) (cadr s))

(define (exponent s) (caddr s))

(define (make-exponentiation b e)
  (cond ((=number? e 0) 1)
	((=number? e 1) b)
	((and (number? b) (number? e)) (** b e))
	(else (list '** b e))))

(define (** b e)
  (exp (* e (log b))))

(define (=number? exp num)
  (and (number? exp) (= exp num)))

(define (sum? x)
  (and (pair? x) (eq? (car x) '+)))

(define (addend s) (cadr s))

(define (augend s) (caddr s))

(define (make-sum a1 a2)
  (cond ((=number? a1 0) a2)
	((=number? a2 0) a1)
	((and (number? a1) (number? a2)) (+ a1 a2))
	(else (list '+ a1 a2))))

(define (product? x)
  (and (pair? x) (eq? (car x) '*)))

(define (multiplier s) (cadr s))

(define (multiplicand s) (caddr s))

(define (make-product m1 m2)
  (cond ((or (=number? m1 0) (=number? m2 0)) 0)
	((=number? m1 1) m2)
	((=number? m2 1) m1)
	((and (number? m1) (number? m2)) (* m1 m2))
	(else (list '* m1 m2))))

;Although variable? exists in guile but it does not equal to symbol?
(define (variable? x) (symbol? x))

(define (same-variable? v1 v2)
  (and (variable? v1) (variable? v2) (eq? v1 v2)))

(define (deriv exp var)
  (cond
   ((number? exp) 0)
   ((variable? exp)
    (if (same-variable? exp var) 1 0))
   ((sum? exp)
    (make-sum (deriv (addend exp) var)
	      (deriv (augend exp) var)))
   ((product? exp)
    (make-sum
     (make-product (multiplier exp)
		   (deriv (multiplicand exp) var))
     (make-product (deriv (multiplier exp) var)
		   (multiplicand exp))))
   ((exponentiation? exp)
	(make-product 
	 (exponent exp)
	 (make-product 
	  (make-exponentiation (base exp) (- (exponent exp) 1))
	  (deriv (base exp) var))))
   (else
    (error "unknown expression type --DERIV" exp))))

(deriv '(** x 3) 'x)

(deriv '(+ (** x 3) (** x 2)) 'x)

;;; Evaluating expression 1 in current module (guile-user)
; => (* 3 (** x 2))

;;; Evaluating expression 2 in current module (guile-user)
; => (+ (* 3 (** x 2)) (* 2 x))
	 